

ForSyDe-Atom

User Manual

George Ungureanu
Department of Electronics
KTH Royal Institute of Technology
ugeorge@kth.se

Contents

1	Introduction	1
1.1	Purpose & organization	1
1.2	Getting FORSYDE-ATOM	2
1.3	Using this document	2
I	Examples & Reports	5
2	Getting Started with FORSYDE-ATOM	7
2.1	Goals	7
2.2	The basics	8
2.3	Toy example: a focus on MoCs	15
2.4	Making your own patterns	29
2.5	Conclusion	32
3	Hybrid CT/DT Models in FORSYDE-ATOM	33
3.1	Goals	33
3.2	RC Oscillator	33
3.3	Conclusion	44
II	API Documentation	45

List of Figures

2.1	Simple process network as composition of processes	9
2.2	Processes structured with 2 layers	11
2.3	Processes structured with 3 layers	11
2.4	Example SY signal plot with Gnuplot	14
2.5	Example SY signal plot with FORSYDE-L ^A T _E X	15
2.6	Views and projections for the toy system	15
2.7	Test input and output signals for a polymorphic process network	30
2.8	Screenshot from online API documentation	31
2.9	Composition of atoms forming the <code>comb</code> pattern	31
3.1	RC orcellator circuit	34
3.2	First FORSYDE-ATOM model of RC oscillator	35
3.3	Test response of the first RC oscillator model	36
3.4	Second FORSYDE-ATOM model of RC oscillator	37
3.5	Test response of the second RC oscillator model	38
3.6	RC filter: circuit and ODE block diagram	39
3.7	FORSYDE-ATOM model of RC filter	40
3.8	Test response of the RC filter model	40
3.9	Third FORSYDE-ATOM model of RC oscillator	41
3.10	Test response of the third RC oscillator model	42
3.11	Execution times and memory consumption for experients	43

Introduction

In this chapter we introduce the purpose, organization and usage of this document, as well as brief instructions and references for helping to set up the FORSYDE-ATOM libraries. The scope is to facilitate the reader's progression through this document.

1.1 Purpose & organization

This book is a living document which gathers material related to FORSYDE-ATOM and binds it in form of a user manual. The vast majority of text contained by this book originates from actual inline or literate source code documentation, in form of examples, tutorials, reports and even library API documentation. This means that this document evolves with the FORSYDE-ATOM project itself and is periodically updated.

FORSYDE-ATOM is a shallow-embedded DSL in the functional programming language Haskell for modeling cyber-physical and parallel systems. It enforces a disciplined way of modeling by separating the manifold concerns of systems into orthogonal *layers*. The FORSYDE-ATOM formal framework aims to providing (where possible) a minimum set of primary common building blocks for each layer called *atoms*, capturing elementary semantics. Even so, the modeling framework provides library blocks and modules commonly used in CPS defined in terms of *patterns* of atoms. For more information on FORSYDE-ATOM itself, please consult the associated scientific publications or the API extended documentation¹.

This document is structured in two parts. The first part gathers examples, tutorials and experiments in a learning progression. Each chapter associated with (and actually generated from) a [Haskell Cabal project](#) included in the `forsyde-atom-examples`² repository. This means that the first part of the book is meant to be read in parallel with running the associated example project which conveniently exports functions to test the listed code “on-the-fly”. The second part of the book is the actual inline API documentation of FORSYDE-ATOM generated with [Haddock](#), which serves also as an extended library report and provides information both on theoretical and implementation issues.

If for any reason you have difficulties following the document or you encounter bugs or discrepancies in the code and text do not hesitate to contact the author(s) or maintainer(s) on GitHub or by email.

¹currently available online at <https://forsyde.github.io/forsyde-atom/>

²available online at <https://github.com/forsyde/forsyde-atom-examples>

1.2 Getting FORSYDE-ATOM

The FORSYDE-ATOM EDSL can be downloaded from <https://github.com/forsyde/forsyde-atom>. The main page contains enough information for acquiring the dependencies and installing the library on your own. However, each example project from the [forsyde-atom-examples](#) repository comes with a set of installation scripts written for user convenience, which should be enough for traversing this manual.

Provided you have an OS installation where the minimum dependencies ([GNU make](#), a [Git CLI client](#), [Haskell Platform](#) and [cabal-install](#)) are working and accessible by your user profile, to run the `getting-started` example associated with chapter 2, you simply need to type in the terminal:

```
# download the examples
git clone https://github.com/forsyde/forsyde-atom-examples.git

# change directory to the desired project folder
cd forsyde-atom-examples/getting-started

# install the project (and dependencies) in a sandbox
make install

# open an interpreter session with the examples loaded
cabal repl
```

1.3 Using this document

DISCLAIMER: the document assumes that the reader is familiar with the syntax of Haskell and the usage of a Haskell interpreter (e.g. `ghci`). Otherwise, we recommend consulting at least the introductory chapters of one of the following books by Lipovača, 2011 and Hutton, 2016 or other recent books in Haskell.

This document has been created using literate programming. This means that all code shown in the listings is compilable and executable. There are two types of code listing found in this document. This style

```
— | API documentation comment
myIdFunc :: a → a
myIdFunc = id
```

shows *source code* as it is found in the implementation files. We have taken the liberty to display some code characters as their literate equivalent (e.g. `->` is shown as `→`, `\` is shown as `λ`, and so on). This style

```
Prelude> 1 + 1
2
```

suggests *interactive commands* given by the user in a terminal or an interpreter session. The listing above shows a typical `ghci` session, where the string after the prompter symbol `>` suggests the user input (e.g. `1 + 1`). Whenever relevant, the expected output is printed one row below that (e.g. `2`).

The code examples are bundled as separate [Cabal](#) packages and is provided as libraries meant to be loaded in an interpreter session in parallel with reading this document. Detailed instructions on how to install the packages can be found in the `README.md` file in each project. The best way to install the packages is within sandboxed environments with all dependencies taken care of, usually scripted within the `make` commands. After a successful installation, to open an interpreter session pre-loaded with the main

sandboxed library, you just need to type in the following command in a terminal from the package root path (the one containing the `.cabal` file):

```
# cabal repl
```

Each section of this document contains a small example written within a library *module*, like:

```
module X where
```

One can access all functions in module `X` by importing it in the interpreter session, unless otherwise noted (e.g. library `X` is re-exported by `Y`).

```
*Y> import X
```

Now suppose that function `myIdFunc` above was defined in module `X`, then one would have direct access to it, e.g.:

```
*Y X> :t myIdFunc
myIdFunc :: a -> a
*Y X> myIdFunc 3
3
```

By all means, the code for `myIdFunc` or any source code for that matter can be copied/-pasted in a custom `.hs` file and compiled or used in any relevant means. The current format was chosen because it is convenient to “get your hands dirty” quickly without thinking of issues associated with compiler suites.

A final tip: if you think that the full name of `X` is polluting your prompt or is hard to use, then you can import it using an alias:

```
*Y> import Extremely.Long.Full.Name.For.X as ShortAlias
*Y ShortAlias>
```

References

- Hutton, Graham (2016). *Programming in Haskell*. 2nd. New York, NY, USA: Cambridge University Press. ISBN: 1316626229, 9781316626221.
- Lipovača, Miran (2011). *Learn You a Haskell for Great Good!: A Beginner's Guide*. 1st. San Francisco, CA, USA: No Starch Press. ISBN: 1593272839, 9781593272838.

Part I

Examples & Reports

Getting Started with FORSYDE-ATOM

This chapter is meant to introduce the reader to using the basic features of FORSYDE-ATOM library as a Haskell EDSL for modeling and simulating embedded and cyber-physical systems. It is not meant to substitute the API documentation nor provide any detail on the theoretical foundation, and it references external documents whenever necessary. It starts from modeling basics, goes through a toy example seen from different perspectives, and into more advanced features like creating custom patterns.

Contents

2.1	Goals	7
2.2	The basics	8
2.2.1	Visualizing your data	12
2.3	Toy example: a focus on MoCs	15
2.3.1	Test input signals	16
2.3.2	SY instance	18
2.3.3	DE instance	20
2.3.4	CT instance	22
2.3.5	SDF instance	24
2.3.6	Polymorphic instance	26
2.4	Making your own patterns	29
2.5	Conclusion	32

2.1 Goals

The main goals of this chapter are:

- introduce the reader to basic modeling features such as: importing library modules, using a Haskell interpreter, using helper functions, composing functions, designing with layers, understanding type signatures, using basic input/output.
- provide a step-by-step guide for modeling a toy system expressing concerns from four layers: function, extended behavior, model of computation and recursive/-parallel composition.
- describe the above system as executing with the semantics dictated by four MoCs: synchronous dataflow (SDF), synchronous (SY), discrete event (DE) and continuous time (CT). For this purpose the system will be first instantiated multiple

times using specialized helpers, and then described as a network of patterns overlaid with MoC semantics by injecting the right data types, thus exposing the polymorphism of atoms.

- briefly introduce the concepts of atoms and patterns and their usage and guide through creating custom patterns and behaviors.

2.2 The basics

This section introduces some basic modeling features of FORSYDE-ATOM, such as helpers and process constructors. The module is re-exported by `AtomExamples.GettingStarted` which is pre-loaded in a `repl` session, so there is no need to import it manually.

```
module AtomExamples.GettingStarted.Basics where
```

We usually start a FORSYDE-ATOM module by importing the `ForSyDe.Atom` library which provides some commonly used types and utilities.

```
import ForSyDe.Atom
```

In this section we will only test [synchronous processes](#) as patterns defined in the `MoC` layer. An extensive library of types, utilities and helpers for SY process constructors can be used by importing the `ForSyDe.Atom.MoC.SY` module.

```
import ForSyDe.Atom.MoC.SY
```

Next we import the `Absent` extended behavior, defined in the `ExB` layer, to get a glimpse of modeling using multiple layers. As with the previous, we need to specifically import the `ForSyDe.Atom.ExB.Absent` library to access the helpers and types.

```
import ForSyDe.Atom.ExB (res11, res21)
import ForSyDe.Atom.ExB.Absent
```

The [signal](#) is the basic data type defined in the `MoC` layer, and it encodes a *tag system* which describes time, causality and other key properties of CPS. In the case of SY MoC, a signal defines a total order between events. There are several ways to instantiate a signal in FORSYDE-ATOM. The most usual one is to create it from a list of values using the `signal` helper. By studying its type signature in the [online API documentation](#), one can see that it needs a list of elements of type `a` as argument, so let us create a test signal `testsig1`:

```
testsig1 = signal [1,2,3,4,5]
```

You can print or check the type of `testsig1`

```
*AtomExamples.GettingStarted> testsig1
{1,2,3,4,5}
*AtomExamples.GettingStarted> :t testsig1
testsig1 :: ForSyDe.Atom.MoC.SY.Core.Signal Integer
```

The type of `testsig1` tells us that the `signal` helper created a SY `Signal` carrying `Integer` values. If you are curious, you can print some information about this mysterious type

```
*AtomExamples.GettingStarted> :info ForSyDe.Atom.MoC.SY.Core.Signal
type ForSyDe.Atom.MoC.SY.Core.Signal a =
  Stream (ForSyDe.Atom.MoC.SY.Core.SY a)
  -- Defined in ForSyDe.Atom.MoC.SY.Core
```

which shows that it is in fact a type alias for a `Stream` of `SY` events. If this became too confusing, please read the `MoC` layer overview in this [online API documentation page](#). Unfortunately the names printed as interactive information are verbose and show

their exact location in the structure of FORSYDE-ATOM. We do not care about this in the source code, since we imported the SY library properly. To benefit from the same treatment in the interpreter session, we need to do the same:

```
*AtomExamples.GettingStarted> import ForSyDe.Atom.MoC.SY as SY
*AtomExamples.GettingStarted SY> :info Signal
type Signal a = Stream (SY a)
-- Defined in ForSyDe.Atom.MoC.SY.Core
```

Another way of creating a SY signal is by means of a `generate` process, which generates an infinite signal from a kernel value. By studying the [online API documentation](#), you can see that the SY library provides a number of helpers for this particular process constructor, the one generating one output signal being `generate1`. Let us first check the type signature for this helper function:

```
*AtomExamples.GettingStarted SY> :t generate1
generate1 :: (b1 -> b1) -> b1 -> Signal b1
```

So basically, as suggested in the [online API documentation](#), this helper takes a "next state" function of type `a -> a`, a kernel value of type `a`, and it generates a signal of type `Signal a`. With this in mind, let us create `testsig2`:

```
testsig2 = generate1 (+1) 0
```

Printing it would jam the terminal... we were serious when we said "infinite"! This is why you need to select a few events from the beginning to see whether the signal generator behaves correctly. To do so, we use the `takeS` utility:

```
*AtomExamples.GettingStarted SY> takeS 10 testsig2
{0,1,2,3,4,5,6,7,8,9}
*AtomExamples.GettingStarted SY> :t testsig2
testsig2 :: Signal Integer
```

`generate` was a process with no inputs. Now let us try a process that takes the two signals `testsig1` and `testsig2` and sums their synchronous events. For this we use the combinatorial process `comb`, and the SY constructor we need, with two inputs and one output is provided by `comb21`. Again, checking the type signature confirms that this is the helper we need:

```
*AtomExamples.GettingStarted SY> :t comb21
comb21 :: (a1 -> a2 -> b1) -> Signal a1 -> Signal a2 -> Signal b1
```

So let us instantiate `testproc1`:

```
testproc1 = comb21 (+)
```

Calling it in the interpreter with `testsig1` and `testsig2` as arguments, it returns:

```
*AtomExamples.GettingStarted SY> testproc1 testsig2 testsig1
{1,3,5,7,9}
```

which is the expected output, as based on the definition of the SY MoC, all events following the sixth one from `testsig2` are not synchronous to any event in `testsig1`.




Figure 2.1: Simple process network as composition of processes

Suppose we want to increment every event of `testsig2` with 1 before summing the two signals. This particular behavior is described by the process network in fig. 2.1. There are multiple ways to instantiate this process network, mainly depending on the coding style of the user:

```

testpn1      = comb21 (+) . comb11 (+1)
testpn2 s2 s1 = testproc1 (comb11 (+1) s2) s1
testpn3 s2    = testproc1 (comb11 (+1) s2)
testpn4 s2 s1 = let s2' = comb11 (+1) s2
               in  testproc1 s2' s1
testpn5 s2    = comb21 (+) s2'
  where s2'   = comb11 (+1) s2

```

All of the above functions are equivalent. `testpn1` uses the point-free notation, i.e. the function composition operator, between two partially-applied process constructors. `testpn2` makes use of the previously-defined `testproc1` to enforce a global hierarchy. `testpn3` is practically the same, but it exposes the partial application mechanism, by not making the `s1` argument explicit. `testpn4` makes use of local hierarchy in form of a `let`-binding, while `testpn5` does so through a `where` clause. Printing them only confirms their equivalence:

```

*AtomExamples.GettingStarted SY> testpn1 testsig2 testsig1
{2,4,6,8,10}
*AtomExamples.GettingStarted SY> testpn2 testsig2 testsig1
{2,4,6,8,10}
*AtomExamples.GettingStarted SY> testpn3 testsig2 testsig1
{2,4,6,8,10}
*AtomExamples.GettingStarted SY> testpn4 testsig2 testsig1
{2,4,6,8,10}
*AtomExamples.GettingStarted SY> testpn5 testsig2 testsig1
{2,4,6,8,10}

```

One key concept of FORSYDE-ATOM is the ability to model different aspects of a system as orthogonal layers. Up until now we only experimented with two layers: the MoC layer, which concerns timing and synchronization issues, and the function layer, which concerns functional aspects, such as arithmetic and data computation. Let us rewind which DSL blocks we have used and group them by which layer they belong:

- the `Integer` values carried by the two test signals (i.e. `0`, `1`, ...) and the arithmetic functions (i.e. `(+)` and `(+1)`) belong to the *function layer*.
- the signal structures for `testsig1` and `testsig2` (i.e. `Signal a`) the utility (i.e. `signal`) and the process constructors (i.e. `generate1`, `comb11` and `comb21`) belong to the *MoC layer*.

It is easy to grasp the concept of layers once you understand how *higher order functions* work, and accept that FORSYDE-ATOM basically relies on the power of functional programming to define structured abstractions. In the previous case entities from the MoC layer “wrap around” entities from the function layer like in fig. 2.2, “lifting” them into the MoC domain. Unfortunately it is not that straightforward to see from the code syntax which entity belongs to which layer. For now, their membership can be determined solely by the user’s knowledge of where each function is defined, i.e. in which module. Later in this guide we will make this apparent from the code syntax, but for now you will have to trust us.

As a last exercise for this section we would like to extend the behavior of the system in fig. 2.1 in order to describe whether the events are happening or not (i.e. are absent or present) and act accordingly. For this, FORSYDE-ATOM defines the *Extended Behavior (ExB) layer*. As suggested in fig. 2.3, this layer extends the pool of values with symbols denoting states which would be impossible to describe using normal values, and associates some default behaviors (e.g. protocols) over these symbols.

The two processes fig. 2.1 are now defined below as `testAp1` and `testAp2`. This time, apart from the functional definition (`name = function`) we specify the type signature as well (`name :: type`), which in the most general case can be considered a




Figure 2.2: Layered structure of the processes in fig. 2.1




Figure 2.3: Layered structure of the processes describing absent events

specification/contract of the interfaces of the newly instantiated component. Both type signatures and function definitions expose the layered structure suggested in fig. 2.3. As specific ExB type, we use `AbstExt` and as behavior pattern constructor we choose a default behavior expressing a resolution `res`.

```
testAp1 :: Num a
=> Signal (AbstExt a) — ^ input signal of absent-extended values
→ Signal (AbstExt a) — ^ output signal of absent-extended values
testAp1 = comb11 (res11 (+1))
```

```
testAp2 :: Num a
=> Signal (AbstExt a) — ^ first input signal of absent-extended values
→ Signal (AbstExt a) — ^ second input signal of absent-extended values
→ Signal (AbstExt a) — ^ output signal of absent-extended values
testAp2 = comb21 (res21 (+))
```

Now all we need is to create some test signals of type `Signal (AbstExt a)`. One way is to use the `signal` utility like for `testAsig1`, but this forces to make use of `AbstExt`'s type constructors. Another way is to use the library-provided process constructor helpers, such as `filter'`, like for `testAsig2`.

```
testAsig1 = signal [Prst 1, Prst 2, Abst, Prst 4, Abst]
testAsig2 = filter' (/=4) testsig2
```

Printing out the test signals in the interpreter session this is what we get:

```
*AtomExamples.GettingStarted SY> testAsig1
{1,2,⊥,4,⊥}
*AtomExamples.GettingStarted SY> takeS 10 testAsig2
{0,1,2,3,⊥,5,6,7,8,9}
```

Trying out `testAp1` on `testAsig1`:

```
*AtomExamples.GettingStarted SY> testAp1 testAsig1
{2,3,⊥,5,⊥}
```

Everything seems all right. Now testing `testAp2` on `testAsig1` and `testAsig2`:

```
*AtomExamples.GettingStarted SY> takeS 10 $ testAp2 testAsig2 testAsig1
{1,3,*** Exception: [ExB.Absent] Illegal occurrence of an absent and present event
```

Uh oh... Actually this *is* the correct behavior of a resolution function for absent events, as defined in synchronous reactive languages such as Lustre Halbwachs et al., 1991. Let us remedy the situation, but this time using another library-provided process constructor, `when'`.

```
testAsig2' = when' mask testsig2
  where
    mask = signal [True, True, False, True, False]
```

Now printing `testAp2` looks better:

```
*AtomExamples.GettingStarted SY> testAsig2'
{0,1,1,3,1}
*AtomExamples.GettingStarted SY> testAp2 testAsig2' testAsig1
{1,3,1,7,1}
```

And recreating the process network from fig. 2.1 gives the expected result:

```
testApn1 = testAp2 . testAp1

*AtomExamples.GettingStarted SY> testApn1 testAsig2' testAsig1
{2,4,1,8,1}
```

This section has provided a crash course in modeling with FORSYDE-ATOM, with focus on a few practical matters, such as using library-provided helpers and constructors and understanding the role of layers. The following sections delve deeper into modeling concepts such as atoms and making use of ad-hoc polymorphism.

2.2.1 Visualizing your data

Up until now, we have made use of the `Show` instance of the FORSYDE-ATOM data types to print out signals on the terminal screen. While this remains the main way to test if a model is working properly, there are alternative ways to plot data. This section introduces the reader to the `ForSyDe.Atom.Utility.Plot` library of utilities for visualizing signals or other data types.

The functions presented in this section are defined in the following module, which is exported by `AtomExamples.GettingStarted`.

```
module AtomExamples.GettingStarted.Plot where
```

We will be using the signals defined in the previous section, so let us import the corresponding module:

```
import AtomExamples.GettingStarted.Basics
```

And, as mentioned, we need to import the library with plotting utilities:

```
import ForSyDe.Atom.Utility.Plot
```

Upon consulting the [API documentation](#) for this module, you might notice that most utilities input a so-called `PlotData` type, which is an alias for a complex structure carrying configuration parameters, type information and data samples. Using Haskell's type classes, FORSYDE-ATOM is able to provide few polymorphic utilities for converting most of the useful types into `PlotData`.

For example, the `prepare` function takes a "plottable" data type (e.g. a signal of values), and a `Config` type, and returns `PlotData`. The `Config` type is merely a record of configuration parameters useful further down in the plotting pipeline. At the time of writing this report¹, a configuration record looked like this:

```
config =
  Cfg { path   = "./fig"      — path where the eventual data files are dumped
      , file   = "plot"      — base name of the eventual files generated
      , rate   = 0.01        — sampling rate if relevant (e.g. ignored by SY signals).
                                — Useful just for e.g. explicit-timed signals.
      , xmax   = 20          — maximum x coordinate. Necessary for infinite signals.
      , labels = ["s1", "s2"] — labels for all signals passed to be plotted.
      , verbose = True       — prints additional messages for each utility.
      , fire   = True        — if relevant, fires a plotting or compiling program.
      , mklatex = True       — if relevant, dumps a LaTeX script loading the plot.
      , mkeps  = True        — if relevant. dumps a PostScript file with the plot.
      , mkpdf  = True        — if relevant, dumps a PDF file with the plot.
      }
```

¹FORSYDE-ATOM v0.2.1

`ForSyDe.Atom.Utility.Plot` provides several of these pre-made configuration objects, which can be modified on-the-fly using Haskell's record syntax, as you will see further on.

Let us see again the signals `testsig1` and `testsig2` defined in the previous section:

```
λ> testsig1
{1,2,3,4,5}
λ> takeS 20 testsig2
{0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19}
```

The utility `showDat` prints out sampled data on the terminal, as pairs of X and Y coordinates:

```
show1 = showDat $ prepare config testsig1
show2 = let cfg = config {xmax=15, labels=["testsig1","testsig2"]}
      in showDat $ prepareL cfg [testsig1,testsig2]
```

```
λ> show1
s1 =
      0  1.0
      1  2.0
      2  3.0
      3  4.0
      4  5.0
<
λ> show2
testsig1 =
      0  1.0
      1  2.0
      2  3.0
      3  4.0
      4  5.0

testsig2 =
      0  0.0
      1  1.0
      2  2.0
      3  3.0
      4  4.0
      5  5.0
      6  6.0
      7  7.0
      8  8.0
      9  9.0
     10 10.0
     11 11.0
     12 12.0
     13 13.0
     14 14.0
```

The function `dumpDat` dumps the data files in a path specified by the configuration object. Based on the `config` object instantiated earlier, after calling the following function you should see a new folder called `fig` in the current path, with two new `.dat` files.

```
dump2 = let cfg = config {xmax=15, labels=["testsig1","testsig2"]}
      in dumpDat $ prepareL cfg [testsig1,testsig2]
```

```
λ> dump2
Dumped testsig1, testsig2 in ./fig
["./fig/testsig1.dat", "./fig/testsig2.dat"]
```

The plot library also has a few functions which create and (optionally) fire `Gnuplot` scripts. In order to make use of them, you need to install the dependencies mentioned in the [API documentation](#). For example, using the function `plotGnu` creates the following plot:

```
plot2 = let cfg = config {xmax=8, labels=["testsig1","testsig2"]}
        in plotGnu $ prepareL cfg [testsig1,testsig2]
```




Figure 2.4: SY signal in Gnuplot as a impulse plot

Different input data creates different types of plots, as we will see in future sections.

One can also generate \LaTeX code which is meant to be compiled with the [FORSYDE- \$\text{\LaTeX}\$](#) package, more specifically its signal plotting library. Check the [user manual](#) for more details on how to install the dependencies and how to use the library itself. Naturally, there is a function `showLatex` which prints out the command for a signals environment defined in [FORSYDE- \$\text{\LaTeX}\$](#) :

```
latex1 = let cfg = config {xmax=8, labels=["testsig1","testsig2"]}
        in showLatex $ prepareL cfg [testsig1,testsig2]
```

```
> latex1
\begin{signalsSY}[] {8.0}
\signalSY[] {1.0:0,2.0:1,3.0:2,4.0:3,5.0:4}
\signalSY[] {0.0:0,1.0:1,2.0:2,3.0:3,4.0:4,5.0:5,6.0:6,7.0:7}
\end{signalsSY}
<
```

Also, there is a command `plotLatex` for generating a standalone \LaTeX document and, if possible, compiling it with `pdflatex`. For example, calling the following function generates the image from section [2.2.1](#).

```
latex2 = let cfg = config {xmax=8, labels=["testsig1","testsig2"]}
        in plotLatex $ prepareL cfg [testsig1,testsig2]
```

The SY signal plot is nothing spectacular, but wraps the events in a matrix of nodes which can be embedded into a more complex \LaTeX figure. Other signals produce other plots. Most generated plots will need manual tweaking in order to look good. Check the [user manual](#) on how to customize each plot.

1.0	2.0	3.0	4.0	5.0					
0.0	1.0	2.0	3.0	4.0	5.0	6.0	7.0		

Figure 2.5: SY signal plot in FORSYDE-L^AT_EX as a matrix of nodes

2.3 Toy example: a focus on MoCs

This example has been used as a case study for introducing the new concepts of FORSYDE-ATOM in the paper of Ungureanu and Sander, 2017. It describes the simple system from fig. 2.6b which exposes four layers, structured like in fig. 2.6a. This system is then fed vectors of signals describing different MoCs and its response is observed. In figs. 2.6c to 2.6f some possible projections on the different layers are depicted. For now they are used just as trivia, and you need not bother with them that much.




Figure 2.6: Views and projections for the toy system

This is a synthetic example meant to introduce as many concepts as possible in a short amount of time and, among others, it highlights:

- the power of partial application for creating parameterized structures, such as the

process network for `stage1`.

- alternative designs for the same `toy` system to show the effect of MoCs. First it is instantiated using different process constructor helpers defined for each MoC separately. Afterwards it is written as one single polymorphic instance using MoC layer patterns, overloaded with execution semantics in accordance with the tag system injected into the system.

For the sake of brevity, we also provide the functional description in the language introduced by Ungureanu and Sander, 2017 in eqs. (2.1)–(2.5) and table 2.1. Do not bother much about this notation either, as this exact definition will appear in the code in a more “human readable” form.

$$\begin{aligned} \text{toy} : \langle V \rangle \rightarrow \langle S \rangle \rightarrow S \\ \text{toy} \langle i \rangle \langle s \rangle = \text{when}_M(\Gamma_w \vdash w_B) \circ \text{reduce}_S(r_M) \circ \text{map}_S(p_{C_M}) \langle i \rangle \langle s \rangle \end{aligned} \quad (2.1)$$

where

$$\text{when}_M(\Gamma_w \vdash w_B)(s) = ((\perp \triangleleft) \circ (\Gamma_w \vdash w_B)) \oplus s \quad (2.2)$$

$$r_M(x, y) = \Gamma_r \vdash r_B \oplus (x, y) \quad (2.3)$$

$$\text{map}_S(p_{C_M}) \langle v \rangle \langle s \rangle = p_{C_M} \diamond \langle v \rangle \diamond \langle s \rangle \quad (2.4)$$

$$p_{C_M}(x, y) = \text{moore}_M(\Gamma_{ns} \vdash ns_B, \Gamma_{od} \vdash od_B, x)(y) \quad (2.5)$$

Table 2.1: CONTEXTS, FUNCTIONS AND INITIAL TOKENS FOR THE SYSTEM IN EQ. (2.1)

MoC	$\Gamma_w \vdash$	$w_B(x)$	$\Gamma_r \vdash$	$r_B(x, y)$	$\Gamma_{ns} \vdash$	$ns_B(x, y)$	$\Gamma_{od} \vdash$	$od_B(x)$	$\langle i \rangle = \langle (t, v) \rangle$
SDF ²	$2, 2 \vdash$	$(x_1 < 0, x_2 < 0) \triangleleft$	$(1, 1) \vdash$	$(x_1 + y_1) \triangleleft$	$(1, 2) \vdash$	$(x_1 + y_1 + y_2) \triangleleft$	$1, 1 \vdash$	$x_1 \triangleleft$	$\langle (, -1) \quad (, 1) \quad (, -1) \quad (, 1) \rangle$
SY	\vdash	$(x < 0) \triangleleft$	\vdash	$(x + y) \triangleleft$	\vdash	$(x + y) \triangleleft$	\vdash	$x \triangleleft$	$\langle (, -1) \quad (, 1) \quad (, -1) \quad (, 1) \rangle$
DE	\vdash	$(x < 0) \triangleleft$	\vdash	$(x + y) \triangleleft$	\vdash	$(x + y) \triangleleft$	\vdash	$x \triangleleft$	$\langle (0.5, -1) \quad (1.4, 1) \quad (1.0, -1) \quad (1.4, 1) \rangle$
CT	\vdash	$(x < 0) \triangleleft$	\vdash	$(x + y) \triangleleft$	\vdash	$(x + y) \triangleleft$	\vdash	$x \triangleleft$	$\langle (1, \lambda t \rightarrow -1) \quad (1.4, \lambda t \rightarrow 1) \quad (1, \lambda t \rightarrow -1) \quad (1.4, \lambda t \rightarrow 1) \rangle$

2.3.1 Test input signals

In the following examples we will use a set of test signals defined in the following module, which is also re-exported by `AtomExamples.GettingStarted` (i.e. you don’t need to import it):

```
module AtomExamples.GettingStarted.TestSignals where
```

The test signals need to define tag systems belonging to different MoCs. Each MoC has an own dedicated module under `ForSyDe.Atom.MoC` which defines atoms, patterns, types and utilities. Just like in the previous section, we need to import the needed modules. This time we name them using short aliases, to disambiguate between the different DSL items, often sharing the same name, but defined in different places.

```
import ForSyDe.Atom.ExB.Absent (AbstExt(...))
import ForSyDe.Atom.MoC.SY      as SY
import ForSyDe.Atom.MoC.DE      as DE
import ForSyDe.Atom.MoC.CT      as CT
```

² $\Gamma_{\text{SDF}} = (\text{consumption rate for first input}, [\text{consumption rate for second input}], \text{production rate})$

```
import ForSyDe.Atom.MoC.SDF      as SDF
import ForSyDe.Atom.MoC.Time     as T
import ForSyDe.Atom.MoC.TimeStamp as Ts
import ForSyDe.Atom.Skeleton.Vector as V
import ForSyDe.Atom.Utility.Plot
```

Let the signals `sdf1`–`sdf4` denote four **SDF** signals, i.e. sequences of events. Instead of using the `signal` utility, we use `readSignal` which reads a string, tokenizes it and converts it to a SDF signal. This utility function needs to be "steered" into deciding which data type to output so in order to specify the type signature we use the inline Haskell syntax `name = definition :: type`. All events, although extended, are present.

```
sdf1 = SDF.readSignal "{ 1, 1, 1, 1, 1, 1}" :: SDF.Signal (AbstExt Int)
sdf2 = SDF.readSignal "{-1, 1,-1, 1,-1, 1}" :: SDF.Signal (AbstExt Int)
sdf3 = SDF.readSignal "{ 0, 0, 1, 1, 0  }" :: SDF.Signal (AbstExt Int)
sdf4 = SDF.readSignal "{-1,-1,-1,-1,-1  }" :: SDF.Signal (AbstExt Int)
```

Similarly, let the signals `sy1`–`sy4` denote four **SY** signals, i.e. all events are synchronized with each other. We use the SY version of `readSignal`, also "steered" by declaring the types inline, and all events are also present.

```
sy1 = SY.readSignal "{ 1, 1, 1, 1, 1, 1}" :: SY.Signal (AbstExt Int)
sy2 = SY.readSignal "{-1, 1,-1, 1,-1, 1}" :: SY.Signal (AbstExt Int)
sy3 = SY.readSignal "{ 0, 0, 1, 1, 0  }" :: SY.Signal (AbstExt Int)
sy4 = SY.readSignal "{-1,-1,-1,-1,-1  }" :: SY.Signal (AbstExt Int)
```

For the **DE** signals `de1`–`de4` we need to specify for each event an explicit tag (i.e. timestamp), as required by the DE tag system. For this, the DE version of `readSignal` reads each event using the syntax `value@timestamp`. Needless to say, all events are also present.

```
de1 = DE.readSignal "{ 1@0                                }" :: DE.Signal (AbstExt Int)
de2 = DE.readSignal "{-1@0, 1@0.7,-1@1.4, 1@2.1,-1@2.8, 1@3.5}" :: DE.Signal (AbstExt Int)
de3 = DE.readSignal "{ 0@0,          1@1.4,          0@2.8      }" :: DE.Signal (AbstExt Int)
de4 = DE.readSignal "{-1@0                                }" :: DE.Signal (AbstExt Int)
```

Let `ct1`–`ct4` denote four **CT** signals. As the events in a CT signal are themselves continuous functions of time, we cannot specify them as mere strings, thus we cannot use a `readSignal` utility any more. This time we will use the CT version of the `signal` utility, where each event is specified as a tuple $(\text{timestamp}, f(t))$, and can be considered as a continuous sub-signal. For representing time we use an alias `Time` for `Rational`, defined in the `ForSyDe.Atom.MoC.Time`. This module also contains utility functions of time, such as the constant function `const` or the sine `sin`. We define local functions to wrap the type returned by a CT subsignal into an `AbstExt` type.

```
pconst = T.const . Prst
ct1 = CT.signal [(0,pconst 1)] :: CT.Signal (AbstExt Time)
ct2 = CT.signal [(0,Prst . (\t → T.sin (T.pi * t)))] :: CT.Signal (AbstExt Time)
ct3 = CT.signal [(0,pconst 0),(1.4,pconst 1),(2.8,pconst 0)] :: CT.Signal (AbstExt Time)
ct4 = CT.signal [(0,pconst (-1))] :: CT.Signal (AbstExt Time)
```

Finally, we need to bundle these signals into vectors of signals, to feed into the toy system from fig. 2.6b and eq. (2.1). For this purpose we pass the four signals of each MoC as a list to the `vector` utility.

```
vsdf = V.vector [sdf1, sdf2, sdf3, sdf4] :: V.Vector (SDF.Signal (AbstExt Int))
vsy  = V.vector [sy1, sy2, sy3, sy4] :: V.Vector (SY.Signal (AbstExt Int))
vde  = V.vector [de1, de2, de3, de4] :: V.Vector (DE.Signal (AbstExt Int))
vct  = V.vector [ct1, ct2, ct3, ct4] :: V.Vector (CT.Signal (AbstExt Time))
```

Now we need to create for each MoC the vectors with the initial states for the Moore machines, i.e. $\langle i \rangle$ from table 2.1. For SDF, SY and DE we are also making use of the

fact that the defined data types are [Readable](#). The data types that we need within the vectors are in accordance to the `moore` process constructor helpers defined in each module, so be sure to check the [online API documentation](#) to understand why we need those particular types. For example, SDF requires a partition (list) of values as initial states whereas DE apart from a value it requires also the duration of the first event. For the CT vector, as before, we cannot read functions as string, so we use the [vector](#) utility. `ict` also shows the usage of the [milisec](#) utility which converts an integer into a timestamp.

```
isdf = read "<[-1],[ 1],[-1],[ 1]>" :: V.Vector [AbstExt Int]
isy  = read "<(-1), 1, (-1), 1>" :: V.Vector (AbstExt Int)
ide  = read "<(1,-1),(1.4, 1),(1.0,-1),(1.4, 1)>"
      :: V.Vector (TimeStamp, AbstExt Int)

ict = V.vector [
  (Ts.milisec 1000, pconst (-1)),
  (Ts.milisec 1400, pconst 1 ),
  (Ts.milisec 1000, pconst (-1)),
  (Ts.milisec 1400, pconst 1 )] :: V.Vector (TimeStamp, Time → AbstExt Time)
```

For the polymorphic instance in section 2.3.6 we need to provide the initial states as wrapped in signals, so we create unit signals:

```
sisdf = SDF.signal <$> isdf
sisy  = SY.unit   <$> isy
side  = DE.unit   <$> ide
sict  = CT.unit   <$> ict
```

Finally, let us instantiate some plotting utilities, to test the different signals throughout the experiments:

```
plot  until lbls = plotGnu . prepare defaultCfg {xmax=until, labels=lbls, rate=0.01}
latex until lbls = plotLatex . prepare defaultCfg {xmax=until, labels=lbls, rate=0.01}
plotV  until lbls = plotGnu . prepareV defaultCfg {xmax=until, labels=lbls, rate=0.01}
latexV until lbls = plotLatex . prepareV defaultCfg {xmax=until, labels=lbls, rate=0.01}
```

2.3.2 SY instance

The SY instance of the toy system is created using process constructor helpers defined in `ForSyDe.Atom.MoC.SY`, and is defined in the following module (re-exported by `AtomExamples.GettingStarted`).

```
module AtomExamples.GettingStarted.SY where
```

As in the previous examples we import the modules we need, and use aliases for referencing them in the code.

```
import ForSyDe.Atom
import ForSyDe.Atom.ExB.Absent (AbstExt)
import ForSyDe.Atom.ExB       as ExB
import ForSyDe.Atom.MoC.SY    as SY
import ForSyDe.Atom.Skeleton.Vector as V
```

Although Haskell's type engine can infer these type signatures, for the sake of documenting the interfaces for each stage, we will explicitly write their types. First, `stage1` is defined as a [farm](#) network of [moore](#) processes, where the initial states are provided by a vector. Its definition makes use of partial application (i.e. arguments which are not explicitly written are supposed to be the same on the LHS as on the RHS). It is defined hierarchically, making use of local name bindings after the `where` keyword.

```
stage1SY :: V.Vector (AbstExt Int)      — ^ vector of initial states
  → V.Vector (SY.Signal (AbstExt Int)) — ^ vector of input signals
  → V.Vector (SY.Signal (AbstExt Int)) — ^ vector of output signals
stage1SY = V.farm21 pcSY
```



```

where
  pcSY = SY.moore11 ns od
  ns   = ExB.res21 (+)
  od   = ExB.res11 id

```

Let us print and plot the inputs against the outputs, using the test signals and plotting functions `latexV` and `plotV` defined in section 2.3.1:

```

λ> isy
<-1,1,-1,1>
λ> vsy
<{1,1,1,1,1},{-1,1,-1,1,-1},{0,0,1,1,0},{-1,-1,-1,-1,-1}>
λ> stage1SY isy vsy
<{-1,0,1,2,3,4,5},{1,0,1,0,1,0,1},{-1,-1,-1,0,1,1},{1,0,-1,-2,-3,-4}>
λ> let latexIn = latexV 6 ["sy1","sy2","sy3","sy4"] vsy
λ> let latexS1 = latexV 7 ["sy1-1","sy2-1","sy3-1","sy4-1"] $ stage1SY isy vsy
λ> let gnuIn = plotV 6 ["sy1","sy2","sy3","sy4"] vsy
λ> let gnuS1 = plotV 7 ["sy1-1","sy2-1","sy3-1","sy4-1"] $ stage1SY isy vsy

```

```

1  1  1  1  1  1
-1 1 -1 1 -1 1
0  0  1  1  0
-1 -1 -1 -1 -1



```

(a) `latexIn`

```

-1  0  1  2  3  4  5
1  0  1  0  1  0  1
-1 -1 -1  0  1  1
1  0 -1 -2 -3 -4

```

(b) `latexS1`(c) `gnuIn`(d) `gnuS1`

The second stage of the toy system in fig. 2.6b is defined as a `reduce` network of `comb` processes. As seen in its type signature, it inputs a vector of signals and it reduces it to a single signal.

```

stage2SY :: V.Vector (SY.Signal (AbstExt Int))
          -> SY.Signal (AbstExt Int)
stage2SY = V.reduce rSY
where
  rSY = SY.comb21 (ExB.res21 (+))

```

Again, let us print and plot the output signals using the test inputs and utilities defined in section 2.3.1.

```


λ> let s2out = (stage2SY . stage1SY isy) vsy
λ> s2out
{0,-1,0,0,2,1}
λ> let latexS2 = latexV 7 ["sy1-2","sy2-2","sy3-2","sy4-2"] s2out
λ> let gnuS2 = plotV 7 ["sy1-2","sy2-2","sy3-2","sy4-2"] s2out

```

Finally, the last stage of the toy system applies a `filter` pattern on the reduced signal to mark all values less than 0 as absent.

0 -1 0 0 2 1

(a) latexS2



(b) gnuS2


```
stage3SY :: SY.Signal (AbstExt Int)
           → SY.Signal (AbstExt Int)
stage3SY = SY.filter (>=0)
>
toySY :: V.Vector (AbstExt Int)           — ^ initial states
       → V.Vector (SY.Signal (AbstExt Int)) — ^ input
       → SY.Signal (AbstExt Int)         — ^ output
toySY i = stage3SY . stage2SY . stage1SY i
```

We print and plot the system response to the test signals defined in section 2.3.1.

```
λ> toySY isy vsy
{0,⊥,0,0,2,1}
λ> let latexS3 = latex 6 ["sy1-3"] $ toySY isy vsy
λ> let gnuS3   = plot 6 ["sy1-3"] $ toySY isy vsy
```

0 ⊥ 0 0 2 1

(a) latexS3



(b) gnuS3

2.3.3 DE instance

The DE instance of the `toy` looks exactly the same as the SY instance in section 2.3.2, but is created using constructors from the `ForSyDe.Atom.MoC.DE` module. This is why we will skip most of the description, and jump straight to testing it. The following file, as you are used to by now, is re-exported by `AtomExamples.GettingStarted`.

```
module AtomExamples.GettingStarted.DE where
```

As previously, we use aliases for the imported modules.

```
import ForSyDe.Atom
import ForSyDe.Atom.ExB.Absent (AbstExt)
import ForSyDe.Atom.ExB       as ExB
import ForSyDe.Atom.MoC.DE    as DE
import ForSyDe.Atom.Skeleton.Vector as V
```

Again, we make the type signatures explicit for documentation purpose. For `stage1` we use the same `farm` network but now using DE `moore` processes.

```

stage1DE :: V.Vector (TimeStamp, AbstExt Int) -- ^ vector of initial states
        -> V.Vector (DE.Signal (AbstExt Int)) -- ^ vector of input signals
        -> V.Vector (DE.Signal (AbstExt Int)) -- ^ vector of output signals
stage1DE = V.farm21 pcDE
  where
    pcDE = DE.moore11 ns od
    ns   = ExB.res21 (+)
    od   = ExB.res11 id


```

When printing `ide` and `vde` we can see the effects of rounding the input floating point numbers to the nearest discrete timestamp. We also have to take into account that the [DE version](#) of the Moore machine produces infinite signals when we print them out. Notice that the generated graphs may need to be tweaked in order to show the information properly.


```

λ> ide
<(1s,-1),(1.3999999999999999s,1),(1s,-1),(1.3999999999999999s,1)>
λ> vde
<{ 1 @0s},{ -1 @0s, 1 @0.6999999999999999s, -1 @1.3999999999999999s, 1 @2.1s, -1 @2.7999999999999999s, 1 @3.5s},{ 0 @0s, 1 @1.3999999999999999s, 0 @2.7999999999999999s},{ -1 @0s}>
λ> fmap (takeS 6) $ stage1DE ide vde
<{ -1 @0s, 0 @1s, 1 @2s, 2 @3s, 3 @4s, 4 @5s},{ 1 @0s, 0 @1.3999999999999999s, 2 @2.0999999999999998s, -1 @2.7999999999999998s, 1 @3.4999999999999997s, 3 @3.4999999999999999s},{ -1 @0s, -1 @1s, -1 @2s, 0 @2.3999999999999999s, 0 @3s, 1 @3.3999999999999999s},{ 1 @0s, 0 @1.3999999999999999s, -1 @2.7999999999999998s, -2 @4.1999999999999997s, -3 @5.5999999999999996s, -4 @6.9999999999999995s}>
λ> let latexIn = latexV 3.3 ["de1","de2","de3","de4"] vde
λ> let latexS1 = latexV 3.3 ["de1-1","de2-1","de3-1","de4-1"] $ stage1DE ide vde
λ> let gnuIn = plotV 3.3 ["de1","de2","de3","de4"] vde
λ> let gnuS1 = plotV 3.3 ["de1-1","de2-1","de3-1","de4-1"] $ stage1DE ide vde


```




(a) latexIn



(b) latexS1



(c) gnuIn



(d) gnuS1

The second stage according to [fig. 2.6b](#) is also defined as a [reduce](#) network but this time we use the DE process constructor for the [comb](#) processes.

```

stage2DE :: V.Vector (DE.Signal (AbstExt Int))
        -> DE.Signal (AbstExt Int)
stage2DE = V.reduce rDE
  where
    rDE = DE.comb21 (ExB.res21 (+))


```

```


λ> let s2out = (stage2DE . stage1DE ide) vde
λ> takeS 10 s2out

```

```
{ 0 @0s, 1 @1s, -1 @1.3999999999999999s, 0 @2s, 2 @2.0999999999999998s, 3 @2.3999999999999999s, -1 @2.7999999999999998s, 0 @3s, 1 @3.3999999999999999s, 3 @3.4999999999999997s}
λ> let latexS2 = latex 3.3 ["de1-2", "de2-2", "de3-2", "de4-2"] s2out
λ> let gnuS2 = plot 3.3 ["de1-2", "de2-2", "de3-2", "de4-2"] s2out
```



(a) latexS2




(b) gnuS2

For the last stage of the toy system there is no DE process constructor in `ForSyDe.Atom.MoC.DE` so we need to create it ourselves.


```
stage3DE :: DE.Signal (AbstExt Int)
          → DE.Signal (AbstExt Int)
stage3DE = deFilter (>=0)
  where
    deFilter p s = DE.comb21 ExB.filter (predSig p s) s
    predSig p s = DE.comb11 (ExB.res11 p) s

toyDE :: V.Vector (TimeStamp, AbstExt Int) — ^ initial states
      → V.Vector (DE.Signal (AbstExt Int)) — ^ input
      → DE.Signal (AbstExt Int) — ^ output
toyDE i = stage3DE . stage2DE . stage1DE i
```

```
λ> takeS 10 $ toyDE ide vde
{ 0 @0s, 1 @1s, ⊥ @1.3999999999999999s, 0 @2s, 2 @2.0999999999999998s, 3 @2.3999999999999999s, ⊥ @2.7999999999999998s, 0 @3s, 1 @3.3999999999999999s, 3 @3.4999999999999997s}
λ> let latexS3 = latex 3.3 ["de1-3"] $ toyDE ide vde
λ> let gnuS3 = plot 3.3 ["de1-3"] $ toyDE ide vde
```



(a) latexS3



(b) gnuS3

2.3.4 CT instance

The CT instance of the toy looks exactly the same as the previous ones in sections 2.3.2 and 2.3.3, but is created using constructors from the `ForSyCt.Atom.MoC.CT` module. The following file, as you are used to by now, is re-exported by `AtomExamples.GettingStarted`.

```
module AtomExamples.GettingStarted.CT where
```

As previously, we use aliases for the imported modules.


```
import ForSyDe.Atom
import ForSyDe.Atom.ExB.Absent (AbstExt)
import ForSyDe.Atom.ExB       as ExB
import ForSyDe.Atom.MoC.CT    as CT
import ForSyDe.Atom.Skeleton.Vector as V
```

Again, `stage1` is a `farm` network of CT `moore` processes.


```
stage1CT :: V.Vector (TimeStamp, Time → AbstExt Time)  — ^ vector of initial states
          → V.Vector (CT.Signal (AbstExt Time))         — ^ vector of input signals
          → V.Vector (CT.Signal (AbstExt Time))         — ^ vector of output signals
stage1CT = V.farm21 pcCT
  where
    pcCT = CT.moore11 ns od
    ns   = ExB.res21 (+)
    od   = ExB.res11 id
```

We cannot print `ict` nor `vct` any more, but we can plot them. Again, the generated plots might need to be tweaked.


```
> let latexIn = latexV 3.5 ["ct1","ct2","ct3","ct4"] vct
> let latexS1 = latexV 3.5 ["ct1-1","ct2-1","ct3-1","ct4-1"] $ stage1CT ict vct
> let gnuIn = plotV 3.5 ["ct1","ct2","ct3","ct4"] vct
> let gnuS1 = plotV 3.5 ["ct1-1","ct2-1","ct3-1","ct4-1"] $ stage1CT ict vct
```




(a) latexIn



(b) latexS1



(c) gnuIn




(d) gnuS1

The second stage is a `reduce` network of CT `comb` processes.


```
stage2CT :: V.Vector (CT.Signal (AbstExt Time))
          → CT.Signal (AbstExt Time)
```

```
stage2CT = V.reduce rCT
  where
    rCT = CT.comb21 (ExB.res21 (+))
```

```
λ> let s2out = (stage2CT . stage1CT ict) vct
λ> let latexS2 = latex 3.3 ["ct1-2","ct2-2","ct3-2","ct4-2"] s2out
λ> let gnuS2 = plot 3.3 ["ct1-2","ct2-2","ct3-2","ct4-2"] s2out
```



(a) latexS2




(b) gnuS2

For the last stage of the toy system there is no CT process constructor in `ForSyDe.Atom.MoC.CT` so we need to create it ourselves.


```
stage3CT :: CT.Signal (AbstExt Time)
  → CT.Signal (AbstExt Time)
stage3CT = ctFilter (>=0)
  where
    ctFilter p s = CT.comb21 ExB.filter (predSig p s) s
    predSig p s = CT.comb11 (ExB.res11 p) s

toyCT :: V.Vector (TimeStamp, Time → AbstExt Time) — ^ initial states
  → V.Vector (CT.Signal (AbstExt Time)) — ^ input
  → CT.Signal (AbstExt Time) — ^ output
toyCT i = stage3CT . stage2CT . stage1CT i
```

```
λ> let latexS3 = latex 3.3 ["ct1-3"] $ toyCT ict vct
λ> let gnuS3 = plot 3.3 ["ct1-3"] $ toyCT ict vct
```



(a) latexS3



(b) gnuS3

2.3.5 SDF instance

The SDF instance of the toy system is created using process constructor helpers defined in `ForSdfDe.Atom.MoC.SDF`, and can be found in the following module (re-exported by `AtomExamples.GettingStarted`).

```
module AtomExamples.GettingStarted.SDF where
```

```
import ForSyDe.Atom
import ForSyDe.Atom.ExB.Absent (AbstExt)
import ForSyDe.Atom.ExB       as ExB
import ForSyDe.Atom.MoC.SDF    as SDF
import ForSyDe.Atom.Skeleton.Vector as V
```

`stage1` is defined as a `farm` network of SDF `moore` processes. As SDF Moore processes are in principle graph loops, we take the initial tokens for each loop from a vector of lists of tokens. Also, both next state and output decoder functions are defined over lists of values instead of values, and they need to be provided within a context which describes the *production* and *consumption* rates. Read more about the particularities of SDF in the [API documentation](#).

```
stage1SDF :: V.Vector ([AbstExt Int])      — ^ vector of initial tokens
           → V.Vector (SDF.Signal (AbstExt Int)) — ^ vector of input signals
           → V.Vector (SDF.Signal (AbstExt Int)) — ^ vector of output signals
stage1SDF = V.farm21 pcSDF
  where
    pcSDF = SDF.moore11 ((1,2),1,ns) (1,1,od)
    ns [x1] [y1,y2] = [ExB.res31 (λa b c → a + b + c) x1 y1 y2]
    od [x1]         = [ExB.res11 id x1]
```

Let us print and plot the inputs against the outputs, using the test signals and plotting functions `latexV` and `plotV` defined in section 2.3.1:


```
> isdf
<[-1],[1],[-1],[1]>
> vsdf
<{1,1,1,1,1},{-1,1,-1,1,-1},{0,0,1,1,0},{-1,-1,-1,-1,-1}>
> stage1SDF isdf vsdf
<{-1,1,3,5},{1,1,1,1},{-1,-1,1},{1,-1,-3}>
> let latexIn = latexV 6 ["sdf1","sdf2","sdf3","sdf4"] vsdf
> let latexS1 = latexV 7 ["sdf1-1","sdf2-1","sdf3-1","sdf4-1"] $ stage1SDF isdf vsdf
> let gnuIn = plotV 6 ["sdf1","sdf2","sdf3","sdf4"] vsdf
> let gnuS1 = plotV 7 ["sdf1-1","sdf2-1","sdf3-1","sdf4-1"] $ stage1SDF isdf vsdf
```

1.0	1.0	1.0	1.0	1.0	1.0
-1.0	1.0	-1.0	1.0	-1.0	1.0
0.0	0.0	1.0	1.0	0.0	
-1.0	-1.0	-1.0	-1.0	-1.0	


(a) latexIn

-1.0	1.0	3.0	5.0
1.0	1.0	1.0	1.0
-1.0	-1.0	1.0	
1.0	-1.0	-3.0	

(b) latexS1



(c) gnuIn



(d) gnuS1

`stage2` is again a `reduce` network of `comb` processes. As with `stage1`, we need to provide the production and consumption rates.

```
stage2SDF :: V.Vector (SDF.Signal (AbstExt Int))
           → SDF.Signal (AbstExt Int)
stage2SDF = V.reduce rSDF
  where
```


```
rSDF = SDF.comb21 ((1,1),1,rF)
rF [x1] [y1] = [ExB.res21 (+) x1 y1]
```

Again, let us print and plot the output signals using the test inputs and utilities defined in section 2.3.1.

```
λ> let s2out = (stage2SDF . stage1SDF isdf) vsdf
λ> s2out
{0,0,2}
λ> let latexS2 = latex 3 ["sdf1-2"] s2out
λ> let gnuS2 = plot 3 ["sdf1-2"] s2out
```

0.0 0.0 2.0

(a) latexS2



(b) gnuS2

As for DE and CT instances, a SDF **filter** process does not really make sense in practice, but for the scope of this toy system, we need to instantiate one ourselves.


```
stage3SDF :: SDF.Signal (AbstExt Int)
           → SDF.Signal (AbstExt Int)
stage3SDF = sdfFilter (>=0)
  where
    sdfFilter p s = SDF.comb21 ((2,2),2,filterF) (predSig p s) s
    filterF pl sl = zipWith ExB.filter pl sl
    predSig p s = SDF.comb11 (1,1,fmap (ExB.res11 p)) s
```

```
>
toySDF :: V.Vector ([AbstExt Int])      — ^ initial tokens
       → V.Vector (SDF.Signal (AbstExt Int)) — ^ input
       → SDF.Signal (AbstExt Int)       — ^ output
toySDF i = stage3SDF . stage2SDF . stage1SDF i
```

```
λ> toySDF isdf vsdf
{0,0}
λ> let latexS3 = latex 3 ["sdf1-3"] $ toySDF isdf vsdf
λ> let gnuS3 = plot 3 ["sdf1-3"] $ toySDF isdf vsdf
```

0.0 0.0

(a) latexS3



(b) gnuS3

2.3.6 Polymorphic instance

In the previous section you've seen how to model systems in FORSYDE-ATOM using the helper functions for instantiating process constructors in different MoCs. In this section we will be instantiating the "raw" polymorphic form of the same process constructors,

not overloaded with any execution semantics. The execution semantics are deduced from the tag system of the input signals, i.e. their types. These process constructors are defined as patterns of MoC atoms in the `ForSdfDe.Atom.MoC` module. The code below is exported by `AtomExamples.GettingStarted`.

```
module AtomExamples.GettingStarted.Polymorphic where
```

Notice that apart from the polymorphic MoC patterns, we are also using "raw" extended behavior and skeleton patterns.

```
import ForSyDe.Atom
import ForSyDe.Atom.MoC.SDF (Prod, Cons)
import ForSyDe.Atom.ExB      as ExB
import ForSyDe.Atom.MoC      as MoC
import ForSyDe.Atom.Skeleton as Skel
```

`stage1` is defined, like in all previous instances, as a `farm` network of `moore` processes.

```
stage1 :: (Skeleton s, MoC m, ExB b)
=> Fun m (b a) (Fun m (b a) (Ret m (b a))) — ^ next state function
→ Fun m (b a) (Ret m (b a)) — ^ output decoder function
→ s (Stream (m (b a))) — ^ signals with initial tokens
→ s (Stream (m (b a))) — ^ vector of input signals
→ s (Stream (m (b a))) — ^ vector of output signals
stage1 ns od = Skel.farm21 (MoC.moore11 ns od)
```

We can immediately observe some main differences in the type signature. First, the `Vector`, `Signal` and `AbstExt` data types are not explicit any more, but suggested as type constraints. The first line in the type signature (`Skel s, MoC m, ExB b`) suggests that the type of `s` should belong to the skeleton layer, the type of `m` should belong to the MoC layer and the type of `b` should belong to the extended behavior layer. Another peculiarity is the presence of the first two structures, but there should be nothing frightening about them: e.g. a structure `Fun m a (Fun m b (Ret m c))` simply stands for the type of a function `a -> b -> c`, which was wrapped in a context specific to a MoC `m`. Read the MoC layer's [API documentation](#) for more on function contexts. This means that the functions for the next state decoder and the output decoder need to be provided as arguments for `stage1`, and they might differ depending on the MoCs. A third peculiarity is that the initial states are provided as signals and not through some specific structure any more. Indeed, the `MoC atoms` extract initial states from signals, and deal with them in different ways depending on the MoC they implement.

The main two classes of MoCs, based on their notion of tags, but also based on how they deal with events, are *timed* MoCs (e.g. SY, DE, CT) and *untimed* MoCs (e.g. SDF). Concerning the functions they lift from layers below, we can say that in FORSYDE-ATOM timed MoCs lift functions on individual values, whereas untimed MoCs lift functions on lists of values (i.e. multiple tokens). Based on this observation, let us define the next state and output decoders for timed and untimed/SDF MoCs.

```
nsT :: (ExB b, Num a) => b a → b a → b a
odT :: (ExB b, Num a) => b a → b a
nsT = ExB.res21 (+)
odT = ExB.res11 id

nsSDF :: (ExB b, Num a) => (Cons, [b a] → (Cons, [b a] → (Prod, [b a])))
odSDF :: (ExB b, Num a) => (Cons, [b a] → (Prod, [b a]))
nsSDF = MoC.ctxt21 (1,2) 1 (λ[x1] [y1,y2] → [ExB.res31 (λa b c → a + b + c) x1 y1 y2])
odSDF = MoC.ctxt11 1 1 (λ[x1] → [ExB.res11 id x1])
```

OBS: for the sake of simplicity, the `ExB` component has been left as part of the `nsT` and `odT`, respectively `nsSDF` and `odSDF`, and not part of `stage1`. Describing all layers within the `stage1` function would have rendered the type signature a bit more complicated and is left as an exercise for the reader.

We postpone plotting the input and output signals for later. Carrying on with instantiating `stage2` as a `reduce` network of `comb` processes:

```
stage2 :: (Skeleton s, MoC m, ExB b)
  => Fun m (b a) (Fun m (b a) (Ret m (b a))) — ^ reduce function
  → s (Stream (m (b a))) — ^ vector of input signals
  → Stream (m (b a)) — ^ output signal
stage2 r = Skel.reduce (MoC.comb21 r)
```

Again, the passed functions need to be specifically defined for each MoC, and for simplicity we include the ExB part as well:

```
rT :: (ExB b, Num a) => b a → b a → b a
rT = ExB.res21 (+)

rSDF :: (ExB b, Num a) => (Cons, [b a] → (Cons, [b a] → (Prod, [b a])))
rSDF = MoC.ctx21 (1,1) 1 (λ[x1] [y1] → [ExB.res21 (+) x1 y1])
```

Finally `stage3`, the `filter` pattern, we create it ourselves in terms of existing ones. This time we incorporate the extended behaviors in the definition of `stage3`, and we only ask for a context wrapper as input argument. Don't be alarmed by the scary type signature, the actual implementation is quite elegant.

```
stage3 :: (MoC m, ExB b, Ord a, Num a)
  => ((b a → b a)
    → Fun m (b a) (Ret m (b a))) — ^ context wrapper for the filter behavior
    → Stream (m (b a)) — ^ input signal
    → Stream (m (b a)) — ^ output signal
stage3 fctx = MoC.comb11 (fctx filtF)
  where filtF a = ExB.filter (ExB.res11 (>=0) a) a
```

And now for the timed/untimed context wrappers:

```
fctxT = id
fctxSDF f = MoC.ctx11 2 2 (fmap f)
```

The full definition of the toy system:

```
toy ns od r fctx is = stage3 fctx . stage2 r . stage1 ns od is
```

And that's it! Let us plot now the test signals and the responses of the system for each stage. This time we will use only \LaTeX plots. We can also plot initial states as they are wrapped as signals. The test results can be seen in fig. 2.7.

```
λ> let noLabel = ["", "", "", ""]
λ> let iSDF = latexV 2 noLabel sisdf
λ> let iSY = latexV 2 noLabel sisy
λ> let iDE = latexV 2 noLabel side
λ> let iCT = latexV 2 noLabel sict
λ>
λ> let vSDF = latexV 6 noLabel vsdf
λ> let vSY = latexV 6 noLabel vsy
λ> let vDE = latexV 3.3 noLabel vde
λ> let vCT = latexV 3.3 noLabel vct
λ>
λ> let s1SDF = latexV 6 noLabel $ stage1 nsSDF odSDF sisdf vsdf
λ> let s1SY = latexV 6 noLabel $ stage1 nsT odT sisy vsy
λ> let s1DE = latexV 3.3 noLabel $ stage1 nsT odT side vde
λ> let s1CT = latexV 3.3 noLabel $ stage1 nsT odT sict vct
λ>
λ> let s2 ns od r is = stage2 r . stage1 ns od is
λ> let s2SDF = latex 6 noLabel $ s2 nsSDF odSDF rSDF sisdf vsdf
λ> let s2SY = latex 6 noLabel $ s2 nsT odT rT sisy vsy
λ> let s2DE = latex 3.3 noLabel $ s2 nsT odT rT side vde
λ> let s2CT = latex 3.3 noLabel $ s2 nsT odT rT sict vct
λ>
λ> let s3SDF = latex 6 noLabel $ toy nsSDF odSDF rSDF fctxSDF sisdf vsdf
```

```

λ> let s3SY = latex 6 noLabel $ toy nsT odT rT fctxT sisy vsy
λ> let s3DE = latex 3.3 noLabel $ toy nsT odT rT fctxT side vde
λ> let s3CT = latex 3.3 noLabel $ toy nsT odT rT fctxT sict vct

```

As expected, the results in fig. 2.7 are exactly the same as the ones presented in sections 2.3.2 to 2.3.5. In conclusion we have successfully instantiated a MoC-agnostic system, whose execution semantics are inferred according to the input data types. This is possible thanks to the notion of type classes, inferred from the host language Haskell. In this section, instead of MoC-specific helpers, we have used the "raw" process constructors as defined in the `ForSdfDe.Atom.MoC` module as patterns of MoC-layer atoms.

This example, used as a case study by Ungureanu and Sander, 2017, has been focused on the MoC layer. A similar approach based on atom polymorphism could target other layers as well since, as you have seen, all layers are implemented as type classes. At the moment of writing this report the extended behavior layer was represented only by the `AbstExt` type, while the skeleton layer had only `Vector`. Nevertheless, future iterations of FORSYDE-ATOM will describe more types.

2.4 Making your own patterns

The final section of this report introduces the reader to constructing custom patterns in FORSYDE-ATOM. Up until now we have been using patterns which were pre-defined as compositions of atoms. Atoms are primitive, indivisible building blocks capturing the most basic semantics in each layer.

```
{-# LANGUAGE PostfixOperators #-}
```

The code for this section is found in the following module, which is *not* re-exported, i.e. needs to be manually imported.

```
module AtomExamples.GettingStarted.CustomPattern where
```

For this exercise, we will create a custom `comb` pattern with 5 inputs and 3 outputs, as a process constructor in the MoC layer. We will test this pattern with a set of SY and a set of DE input signals, thus we need to import the following modules:

```

import ForSyDe.Atom
import ForSyDe.Atom.MoC
import ForSyDe.Atom.MoC.SY as SY
import ForSyDe.Atom.MoC.DE as DE

```

The best way to start building your own patterns is to study the source code for the existing patterns and see how they are made. If you don't want to dig into the source code of FORSYDE-ATOM, there is a link in the [API documentation](#) for each exported element, as suggested in fig. 2.8.

Studying the `comb22` pattern, you can see that it is defined in terms of the `lift` and `sync` atoms which are represented by the infix operators `-.-` and `-*-` respectively, and the `unzip` utility represented by the postfix operator `-*<`. `lift` and `sync` are atoms because they capture an interface for execution semantics, whereas `unzip` is just a utility because it is merely a type traversal which alters the structure of data types and rebuilds it to describe "signals of events carrying values".

Considering the applicative nature of the `-.-` and `-*-` atoms, the `comb` pattern with 5 inputs and 3 outputs can be written as the mathematical formula below. This one-liner tells that function `f` is "lifted" into the MoC domain, and applied to the five input signals which are synchronized. The `-*<<` postfix operator then "unzips" the resulting signal of triples into three synchronized signals of values. The applicative mechanism explained in the previous paragraph is depicted in fig. 2.9.




Figure 2.7: Inputs and outputs for the polymorphic toy system in section 2.3.6

```

comb22 Source
:: MoC e
=> Fun e a1 (Fun e a2 (Ret e b1, Ret e b2))

```

Figure 2.8: Screenshot from the API documentation. The link to the source code is marked with a red rectangle

```

comb53 f s1 s2 s3 s4 s5
= (f -.- s1 -* s2 -* s3 -* s4 -* s5 -*<<)

```




Figure 2.9: Composition of atoms forming the comb pattern

If we want to restrict the pattern to one specific MoC, then we must mention this in the type signature we associate it with, like in the example below.

```

comb53SY :: (a1 → a2 → a3 → a4 → a5 → (b1,b2,b3))
→ SY.Signal a1    — ^ input signal
→ SY.Signal a2    — ^ input signal
→ SY.Signal a3    — ^ input signal
→ SY.Signal a4    — ^ input signal
→ SY.Signal a5    — ^ input signal
→ ( SY.Signal b1
  , SY.Signal b2
  , SY.Signal b3) — ^ 3 output signals
comb53SY f s1 s2 s3 s4 s5
= (f -.- s1 -* s2 -* s3 -* s4 -* s5 -*<<)

```

To test the output, let us create five signals and a function that needs to be lifted. For the example, the terminal printouts should suffice to test our simple pattern.

```

λ> import AtomExamples.GettingStarted.CustomPattern as CP
λ> let fun a b c d e = (a+c+e, d-b, a*e)
λ> let sy1 = SY.signal [1,2,3,4,5]
λ> let sy2 = SY.comb11 (+10) sy1
λ> let sy3 = SY.constant1 100
λ> let de1 = DE.signal [(0,1),(3,2),(7,3),(9,4),(11,5)]
λ> let de2 = DE.signal [(0,11),(3,12),(5,13),(9,14),(11,15)]
λ> let de4 = DE.constant1 100
λ>
λ> let (o1,o2,o3) = CP.comb53 fun sy1 sy1 sy2 sy2 sy3
λ> o1
λ> {112,114,116,118,120}
λ> o2
λ> {10,10,10,10,10}
λ> o3
λ> {100,200,300,400,500}
λ>
λ> let (o1,o2,o3) = CP.comb53 fun de1 de1 de2 de2 de4
λ> o1
λ> { 112 @0s, 114 @3s, 115 @5s, 116 @7s, 118 @9s, 120 @11s}

```

```

λ> o2
λ> { 10 @0s, 10 @3s, 11 @5s, 10 @7s, 10 @9s, 10 @11s}
λ> o3
λ> { 100 @0s, 200 @3s, 200 @5s, 300 @7s, 400 @9s, 500 @11s}
λ>
λ> let (o1,o2,o3) = CP.comb53SY fun sy1 sy1 sy2 sy2 sy3
λ> o1
λ> {112,114,116,118,120}
λ> o2
λ> {10,10,10,10,10}
λ> o3
λ> {100,200,300,400,500}
λ>
λ> let (o1,o2,o3) = CP.comb53SY fun de1 de1 de2 de2 de4
<interactive>:71:56-58:
  Couldn't match type 'DE Integer' with 'SY b3'
  Expected type: SY.Signal b3
  Actual type: DE.Signal Integer
  Relevant bindings include
    it :: (SY.Signal b3, SY.Signal b2, SY.Signal b3)
          (bound at <interactive>:71:1)
  In the second argument of 'c0omb53SY', namely 'de1'
  In the expression: comb53SY fun de1 de1 de2 de2 de4
...

```

2.5 Conclusion

This report has introduced the reader to the basic features of FORSYDE-ATOM a framework for modeling and testing of cyber-physical systems. It has covered basic usage such as instantiating systems and plotting signals. It briefly went through concepts such as layers, atoms and patterns, and has focused on their practical usage. A step-by-step tutorial has been presented, showing alternative ways to instantiate systems and demonstrating the polymorphism of layers.

The reader is recommended to further consult the [API documentation](#) which also acts as a manual for the library, as well as the related publications listed on the [project web site](#). Future reports will assume familiarity with using and understanding the framework and will focus mainly on results.

References

- Halbwachs, Nicholas et al. (1991). “The synchronous data flow programming language LUSTRE”. In: *Proceedings of the IEEE* 79.9, pp. 1305–1320.
- Ungureanu, George and Ingo Sander (2017). “A layered formal framework for modeling of cyber-physical systems”. In: *2017 Design, Automation & Test in Europe Conference & Exhibition (DATE)*. IEEE, pp. 1715–1720.

Hybrid CT/DT Models in FORSYDE-ATOM

This chapter gathers examples and experiments which involve hybrid models focusing on the semantics and implications of combining and translating between continuous and discrete domains. As such, the focus is mainly on exploring alternative models and analyzing the implications on fidelity, precision and performance. This chapter is also meant to support the associated scientific publications with experimental results.

Contents

3.1 Goals	33
3.2 RC Oscillator	33
3.3 Conclusion	44

3.1 Goals

The reader is assumed to have been familiarized with the FORSYDE-ATOM modeling framework. A good resource for that is chapter 2. The main goals of this chapter are:

- explore alternative FORSYDE-ATOM models for widely known hybrid (discrete and continuous time) systems, with the scope of analyzing the implications from the modeling and simulation perspective.
- train the reader into the decision-making process of modeling hybrid systems, and the trade-offs involved. While as per writing this report the FORSYDE-ATOM modeling framework is still limited, future directions are hinted.
- support the associated scientific publications with the complete experiments and results for the models used as case studies. These publications include: Ungureanu et al., 2018.

3.2 RC Oscillator

In this section we model the RC oscillator setup in fig. 3.1. Despite its apparent simplicity, shows an interesting problem in CPS: how continuous systems react to discrete stimuli. This example was used in Ungureanu et al., 2018, and for a study on the theoretical implications of the chosen models, we strongly recommend reading this paper before going further. The source code for this section is found in the following module:

```
module AtomExamples.Hybrid.RCOsc where
```




Figure 3.1: RC oscillator setup

These are the dependencies that need to be included within this module:

```
import ForSyDe.Atom                — general utilities
import ForSyDe.Atom.MoC.CT        as CT    — CT MoC library
import ForSyDe.Atom.MoC.DE        as DE    — DE MoC library
import ForSyDe.Atom.MoC.SY        as SY    — SY MoC library
import ForSyDe.Atom.MoC.Time      as T      — utilities for functions of time
import ForSyDe.Atom.MoC.TimeStamp (milisec) — utilities for time stamps
import ForSyDe.Atom.Utility.Plot   — plotting utilities
```

The circuit in fig. 3.1 is characterized by two states: 1) sw_1 is closed and sw_2 open, equivalent to the situation where the capacitor C charges with V_{DD} ; 2) sw_1 is open and sw_2 is closed, equivalent to the situation where the capacitor C discharges to the ground through R . For the scope of this example, to keep the model simple and deterministic, we assume that sw_1 and sw_2 open/close alternatively at the same discrete time instants, there are no intermediary transitions, and we ignore the effects of discharge through switches.

Let us analyze case 1) above. The circuit in fig. 3.1 becomes a RC integrator, where the input signal is applied to the resistance with the output taken across the capacitor. The amount of charge that is established across the plates of the capacitor is equal to the time domain integral of the current, like in eq. (3.1). The rate at which the capacitor charges is directly proportional to the amount of the resistance and capacitance giving the time constant of the circuit like in eq. (3.2). As the capacitors current can be expressed as the rate of change of charge, Q with respect to time, we can express the charge at any instant of time like in eq. (3.3). Eq. (3.3) brings together the previous equations and uses the instant voltage charge formula to obtain the final integral formula for V_O .

$$i_C(t) = C \frac{\partial V_C(t)}{\partial t} = \frac{V_{DD}}{R} \quad (3.1)$$

$$RC = R \frac{Q}{V} = R \frac{i \times T}{i \times R} = T \quad (3.2)$$

$$i = \frac{\partial Q}{\partial t} \Rightarrow Q = \int i \, dt \quad (3.3)$$

$$V_O = V_C = \frac{Q}{C} \stackrel{(3)}{=} \frac{1}{C} \int i \, dt \stackrel{(1)}{=} \frac{1}{C} \int \frac{V_{DD}}{R} dt = \frac{1}{RC} \int V_{DD} dt \quad (3.4)$$

If an ideal step voltage pulse is applied, that is with the leading edge and trailing edge considered as being instantaneous, the voltage across the capacitor will increase for charging exponentially over time at a rate determined by eq. (3.5). Similarly, in case 2) the circuit becomes an RC bridge where, assuming that C has been fully charged, it

now discharges to the ground at a rate determined by eq. (3.6).

$$V_{O,\text{charge}}(t) = V_C(t) = V_{DD} \left(1 - e^{-\frac{t}{RC}}\right) \quad (3.5)$$

$$V_{O,\text{discharge}}(t) = V_C(t) = V_{DD} \left(e^{-\frac{t}{RC}}\right) \quad (3.6)$$

Translating eqs. (3.5) and (3.6) into FORSYDE-ATOM Haskell code, we get the following, assuming `t0` is the moment where the switch occurred, and ignoring the factor of V_{DD} (which is considered and scaled in the output decoder anyway).

```

— | RC time constant.
rc = 0.1

— | V_O(t0,t) during the discharging. t0 is the instant when the switch occurred.
vcDischarge t0 = λt → T.exp (-(t-t0) / rc)

— | V_O(t0,t) during the charging. t0 is the instant when the switch occurred.
vcCharge      t0 = λt → 1 - T.exp (-(t-t0) / rc)

```

Naturally, in FORSYDE-ATOM we can model the oscillator in multiple ways, depending on what aspect we want to focus. Our first example models the RC oscillator as a Mealy finite state machine which embeds the continuous and discrete time semantics as defined in the `ForSyCt.Atom.MoC.CT` library, like in fig. 3.2. As such, we imply from the model that the switching of sw_1 and sw_2 is performed periodically after a certain τ , inferred from the initial state of the `CT.mealy` process. As such, the state machine is loaded with the voltage charging rule in eq. (3.5), and a duration τ . The configuration of the `mealy` pattern ensures that at each multiple of τ the next state function `ns` will be performed, negating the state rule (i.e. the rate for V_O), basically translating back and forth between eqs. (3.5) and (3.6). As mentioned before, the output decoder `od` merely rescales V_O with V_{DD} .




Figure 3.2: Internal pattern of the initial RC oscillator setup

```

— | RC oscillator model as CT FSM with discrete semantics inherent in the CT model.
osc1 :: CT.Signal Rational — ^ VDD as input signal
      → CT.Signal Rational — ^ V_O as output signal
osc1 = CT.mealy11 ns od (milisec 500, vcCharge 0)
  where
    ns v _ = 1 + (-1 * v)
    od     = (*)

```

Plotting the output against an "arbitrary" V_{DD} , we get fig. 3.3.

```

— | plotting configuration that will be used throughout this section
cfg = defaultCfg {xmax=3, rate=0.01}

```


```

— | example input for testing 'osc1'
vdd1 = CT.signal [(0,λ_→2),(1,λ_→1.5),(2,λ_→1)] :: CT.Signal Rational

— | plotting the response of 'osc1'
plot1 = plotGnu $ prepareL cfg {labels=["V_{DD}","V_O"]} $ [vdd1, osc1 vdd1]

```

The model for `osc1`, although strikes out as simple and elegant, is by all means limited. In the following paragraphs we will address these limitations one at a time,

Figure 3.3: Response of `osc1`

and try find solutions that overcome them in order to include more realistic or at least a family of behaviors correctly covering classes of (increasing) non-determinism.

Lee, 2016 argues that determinism is an attribute of the model, and it is dependent on what we consider as inputs and outputs. As such, the model `osc1` in fig. 3.2 is a deterministic model for the circuit in fig. 3.1 and the output in fig. 3.3 is the correct response in the following conditions:

- the changes in V_{DD} happen at multiples of τ . If a changes occur at any other time the response will be shown, but it will not describe the RC circuit in a realistic manner, as we will see shortly.
- the time constant follows roughly the rule $5RC \leq \tau$ where the chosen τ is a part of the initial state of the Mealy state machine. This says that the discharging occurs when the capacitor is charged at least 99.3% and vice-versa, according to eqs. (3.5) and (3.6).
- $\tau > 0$. $\tau = 0$ would render the system as non-causal and would manifest Zeno behavior. In practical terms, this is the equivalent of the simulation being stuck and not advancing time.

Another property (or limitation, depending on what your intentions are) is that the switching of sw_1 and sw_2 cannot be controlled and is a property of the Mealy machine. This switching is inferred by τ which sets the (discrete) period of these events. Let us change that by giving the possibility to control the state of the capacitor as charging/discharging through a signal of discrete events. A system such as the one we propose implies some subtle changes in the modelling which require a deeper understanding on the underlying MoCs. First of all, we still require a stateful process (a state machine) which captures the notion of working modes, but which reacts to a state change instantaneously. But this implies that $\tau = 0$ which, as said above, would lead to Zeno

behavior. On the other hand, the particular behavior required is described precisely by *synchronous reactive* MoC, where the response of a system is performed in "zero time".

Continuing to adhere to the school of thought of Lee, 2016, where the merit of the modeler lies in choosing the right modeling paradigm for the right problem¹, we choose to model the above described system as a synchronous reactive (SY) state machine wrapped inside a DE/CT environment like in fig. 3.4. As such, we should note a few particularities of this model:

- in CT the carried values are implicit functions of time, i.e. they carry time semantics. In DE and SY, the time semantics make no sense, thus the same functions of time are explicit (where **Time** itself is just a data type $\in V$). Therefore when converting $\mathcal{S}_{CT}(\alpha) \mapsto \mathcal{S}_{DE}(t \rightarrow \alpha)$ and vice-versa $\mathcal{S}_{DE}(t \rightarrow \alpha) \mapsto \mathcal{S}_{CT}(\alpha)$.
- in SY tags are useless, therefore DE tags and values are split into two separate SY signals upon conversion. This way we can easily recreate the DE signal without loss of information. So upon conversion $\mathcal{S}_{DE}(\alpha) \mapsto \mathcal{S}_{SY}(t) \times \mathcal{S}_{SY}(\alpha)$ and vice-versa $\mathcal{S}_{SY}(t) \times \mathcal{S}_{SY}(\alpha) \mapsto \mathcal{S}_{DE}(\alpha)$.
- the discrete control signal S_{control} , carries nothing, and it is used only for the timestamps generated by its tags, i.e. for generating $\mathcal{S}_{SY}(t)$. These timestamps are further used by the **od** function to determine the time when the switch occurred **t0** in the formulas for V_O in eqs. (3.5) and (3.6).




Figure 3.4: RC oscillator model which reacts to a discrete control signal. Shapes decorating signals suggest the type of carried tokens: circles = scalars; squares = functions

Like in the model from fig. 3.2, the model for **osc2** in fig. 3.4 scales the output with V_{DD} through a combinational CT process. The next state decoder function **ns** does not manipulate the rule for V_C anymore, but rather switches between the states **Charge** and **Discharge**, and the output decoder **od** selects the proper V_C rule from eq. (3.5) or eq. (3.6) respectively, based on the current state.

```
— | Encodes the capacitor state
data CState = Charging | Discharging
```

We encode the capacitor state (i.e. the states of sw_1 and sw_2) through the **CState** data type. Thus the code for fig. 3.4 is:

```
— | RC oscillator model as FSM which reacts to discrete impulses
osc2 :: CT.Signal Rational — ^ VDD input signal
      → DE.Signal ()      — ^ control signal of discrete impulses
      → CT.Signal Rational — ^ output signal
osc2 s = CT.comb21 (*) s . embed (SY.mealy11 ns od Charging)
where
  — SY next state function
  ns Charging    = Discharging
  ns Discharging = Charging
  — SY output decoder function
```

¹Lee argues that the cost we pay in the loss of determinism is a property of the chosen modeling framework, and *not* of the model itself.

```

od Charging    t0 = vcCharge (time t0)
od Discharging t0 = vcDischarge (time t0)
— wrapper that embeds a SY process into a mixed DE/CT environment
embed p de     = let (t, _) = DE.toSY de
               in DE.toCT $ SY.toDE t (p t)

```

Now let us create two situations to test `osc2`. The control signal `sCtrl` injects events at timestamps 0, 0.6, 1.4, 2.3 and 2.8, causing the SY state machine to react each time changing its state from `Charge` to `Discharge` and vice-versa. The first V_{DD} input `vdd21` is mirroring an “ideal case”, when changes occur at time instants where switching happens, meaning that it does not affect the evolution of the V_O rule. On the contrary, `vdd22` comes at arbitrary times, thus affecting the output in a way that is unrealistic, i.e. V_O changes values abruptly and instantaneously, which strays away from the acceptable behavior of a capacitor. This can be seen in fig. 3.5.


```

— | Signal of discrete events. Carries nothing, it is used only for the time stamps.
sCtrl = DE.signal [(0,()),(0.6,()),(1.4,()),(2.3,()),(2.8,())] :: DE.Signal ()

— | example input for testing 'osc2'
vdd21 = CT.signal [(0,λ→2),(1.4,λ→1.5),(2.8,λ→1)] :: CT.Signal Rational
vdd22 = CT.signal [(0,λ→1),(0.8,λ→2),(1.6,λ→1.5)] :: CT.Signal Rational

— | plotting the example responses of 'osc2'
plot21 = plotGnu $ prepareL cfg {labels=["V_{DD}","V_O"]} $ [vdd21, osc2 vdd21 sCtrl]
plot22 = plotGnu $ prepareL cfg {labels=["V_{DD}","V_O"]} $ [vdd22, osc2 vdd22 sCtrl]

```

Figure 3.5: Response of `osc2`

The second case in fig. 3.5 can be explained easily if we consider the fact that V_O itself is the result of a statically pre-calculated function of time. There is no notion of feedback response to the continuous inputs, the only feedback is synchronous reactive for `osc2` and even `osc1`. In other words any change on the input will affect how the output “looks like”, but it will not affect the continuous behavior which, as said, is pre-calculated. In order to influence the continuous behavior, some sort of continuous feedback is necessary, which is usually non-causal, thus uncomputable. Without going too much into detail we can say that in order to model a realistic behavior of the capacitor response in V_O , we need to embed an ordinary differential equation (ODE) solver within a synchronous reactive state machine², which models precisely eq. (3.4).

$$\int_{t_0}^{t_0+h} f(t, y(t)) \partial t \approx h f(t_0, y(t_0)) \quad (3.7)$$

$$\dot{y}_n = \frac{hRC}{h + RC} \left(\frac{1}{RC} y_n + \frac{1}{h} y_{n-1} \right) \quad (3.8)$$

²theoretical implications will be analyzed in the upcoming journal publications




Figure 3.6: RC bridge: circuit (left); block diagram based on eq. (3.4) (right)

First let us model a simple RC bridge like in fig. 3.6 which acts like a low-pass filter, and acts according to eq. (3.4). As mentioned, we model this circuit as a SY state machine embedded within a CT environment. Why a state machine? Because the circuit itself, due to the capacitor, is a memory system, i.e. its output is dependent on its history as well as inputs. The previous "history" at the start of simulation is encoded inside the initial state. The next state decoder itself needs to "feed-through" the input and mirror the behavior of the block diagram fig. 3.6. On the other hand, this block diagram shows a non-causal system, which as such is uncomputable, and needs to be transformed into a solvable form. As per the writing of this report, FORSYDE-ATOM³ did not provide generic ODE solvers, thus for didactic purpose we write the following numerical solver for our RC circuit in fig. 3.6 using Euler's trapezoidal method eq. (3.7), by substituting by hand eq. (3.4) with its feed-forward version in eq. (3.8):

```

— | ODE solver for a simple RC low-pass filter using Euler's method
euler :: TimeStamp — ^ time step for the solver precision
  → (Time → Rational) — ^ the input function being integrated
  → Rational — ^ the "history" of the integral at t0
  → TimeStamp — ^ t0
  → Time → Rational — ^ a function of time
euler step f p t0 t = iterate p t0
  where
    h = time step
    — by-hand substitution of eq.(4) using the trapezoidal rule
    calc vp v = (h * rc)/(h + rc) * (1/rc * v + 1/h * vp)
    — loop which calculates the integral step-wise from t0 to t
    iterate st ti
      | t < time ti = st
      | otherwise = iterate (calc st $ f (time ti)) (ti + step)

```

As said earlier, we model the RC circuit in fig. 3.6 by including the `euler` solver into a feed-through state machine (see the `state` pattern) like in fig. 3.7. Like for `osc2`, the timestamps resulted after splitting the tags from values when interfacing between DE and SY are used for determining the t_0 s for each new incoming event. The solver inputs functions of time and outputs functions of time (i.e. the solver itself is the output).

```

rcfilter :: CT.Signal Rational → CT.Signal Rational
rcfilter s
  = let (ts, sy) = DE.toSY $ CT.toDE s
      out       = SY.state21 ns (λ_→0) ts sy
      ns p t s = euler 0.01 s (p (time t)) t
      in DE.toCT $ SY.toDE ts out

```

Testing the filter against a square wave signal, we get the response plotted in fig. 3.8. As can be clearly seen, the behavior is the right one and it reacts correctly to the inputs, taking into account the current state of the system, i.e. its history, and it is a *continuous* signal in the true sense of the word.

```

— | square wave signal for testing 'rcfilter'

```

³version 0.2.2




Figure 3.7: FORSYDE-ATOM model of the RC bridge in fig. 3.6

```
vi3 = CT.signal [(0,λ_→2), (0.3,λ_→0), (1,λ_→1.5), (1.5,λ_→0), (1.7,λ_→1), (2.5,λ_→0)] :: CT.Signal Rational
-- | plotting the example response of 'rcfilter'
plot31 = plotGnu $ prepareL cfg {labels=["V_I","V_0"]} $ [vi3, rcfilter vi3]
```




Figure 3.8: Response of rcfilter

Now let us modify `osc2` from fig. 3.4 to mirror the correct behavior of the circuit in fig. 3.1. Seems to it that there is not much to do, as `rcfilter` already responds correctly to any input. On the other hand the RC oscillator as represented in fig. 3.1 admits as inputs both V_{DD} and the control signal for sw_1 and sw_2 . To correctly model that in FORSYDE-ATOM we need to separate the FSM which controls the state of the capacitor as charging or discharging, which in turn will generate V_I . We do that like in fig. 3.9.

```
-- | ODE-based RC oscillator model
osc4 :: CT.Signal Rational -- ^ VDD input signal
      → DE.Signal ()      -- ^ control signal of discrete impulses
      → CT.Signal Rational -- ^ output signal
osc4 vdd ctl
```

```

= let — generator for the switch state variable
    swState = DE.embedSY11 (SY.stated11 swF Charging) ct1
    swF Charging    = Discharging
    swF Discharging = Charging
    — transforms VDD into VI for the RC filter
    vddSwitched    = DE.comb21 vddF swState $ CT.toDE vdd
    vddF Charging   v = v
    vddF Discharging _ = λ_→0
    — state machine with ODE solver modeling an RC filter
    v0Out           = embed (SY.state21 nsEU (λ_→0)) vddSwitched
    nsEU p t s      = euler 0.01 s (p $ time t) t
    — custom wrapper that embeds a SY tag-aware process into a DE environment
    embed p de      = let (t, v) = DE.toSY de
                        in SY.toDE t (p t v)
in DE.toCT v0Out

```




Figure 3.9: FORSYDE-ATOM model of the RC oscillator described by `osc4`

Testing `osc4` against the same inputs as `osc2` in fig. 3.5, we get the response plotted in fig. 3.10, which is now the correct behavior of the RC circuit with respect to its inputs. As expected, the output describes correctly the dynamics of the system based on the state and history of the capacitor.

```

| plot41 = plotGnu $ prepareL cfg {labels=["V_{DD}","V_O"]} $ [vdd22, osc4 vdd22 sCtrl]

```

We have shown three different models of an RC oscillator circuit represented in fig. 3.1 at different levels of complexity, and a model of an RC bridge represented in fig. 3.6. The fact that they are different does not mean that either is “more correct/incorrect” than the other. Either of them might very well be treated as “correct” depending on what we consider or not the acceptable inputs (note that we have not defined them on purpose). As expected, the more we consider the inputs as sources of non-determinism, the more complex our model needs to be in order to cover “special” cases. As you might guess already, this comes at a severe cost of run-time performance.

Let us briefly measure this cost in performance between the four presented models. For this, we consider two situations:

Experiment 1 we need to find out V_O at $t = 2.8$ seconds.

Experiment 2 we need to sample V_O for the whole period $t = [0, 3]$ seconds, with a precision of 50 milliseconds.

We consider the same input scenarios for all 4 models:

```

— | plotting/sampling configuration for performance testing
cfgTest = defaultCfg {xmax=3, rate=0.005}
— | VDD input for performance testing
vddTest = CT.signal [(0,λ_→2),(1,λ_→1.5),(2,λ_→1)] :: CT.Signal Rational
— | VI input for performance testing of the RC filter model

```




Figure 3.10: Response of osc4

```
viTest = CT.signal [(0,λ_→2),(0.5,λ_→0),(1,λ_→1.5),(1.5,λ_→0),(2,λ_→1),(2.5,λ_→0)]
:: CT.Signal Rational
— | Switch control signal for performance testing
ctlTest = DE.signal [(0,()),(0.5,()),(1,()),(1.5,()),(2,()),(2.5,())] :: DE.Signal ()
```

DISCLAIMER: we have measured the performance for all experiments using the `:set +s` directive in a `ghci` interpreter session, which offers some crude information about the run-time. The measurements are by no means realistic for compiled and optimized programs, but rather give a rough idea about the raw, un-optimized functions run directly in the interpreter. In any case, this offers a common ground for comparing implementations between them to observe trends and the influences of the design decisions, but not to judge FORSYDE-ATOM as such.


```
λ> :set +s
λ> fromRational $ osc1 vddTest          'CT.at' 2.8
4.9787192469339395e-2
(0.01 secs, 398,792 bytes)
λ> fromRational $ osc2 vddTest ctlTest 'CT.at' 2.8
4.9787192469339395e-2
(0.01 secs, 385,952 bytes)
λ> fromRational $ rcfilter viTest       'CT.at' 2.8
5.169987618408467e-2
(0.08 secs, 34,044,128 bytes)
λ> fromRational $ osc4 vddTest ctlTest 'CT.at' 2.8
5.169987618408467e-2
(0.07 secs, 34,047,720 bytes)
```

The experimental results on a computer with Intel® Core™ i7-3770 CPU @ 3.40GHz × 8 threads, and 31,4 GiB RAM are shown in table 3.1. As expected, `osc1` and `osc2` perform in almost negligible time for both experiments due to lazy evaluation which performs computation only at the requested evaluation point. However, the lazy evaluation is costly in terms of run-time memory, fact noticed especially for `rcfilter` and `osc4` due to the fact that intermediate structures need to be stored before evaluating.

Model	Experiment 1		Experiment 2	
	time (s)	memory (MB)	time (s)	memory (MB)
osc1	0.01	0.4	0.09	23
osc2	0.01	0.4	0.09	18
rcfilter	0.08	34	5.04	1930
osc4	0.07	34	5.56	1930

Table 3.1: Experimental results for testing the RC models

The complexity of `rcfilter` and `osc4` are seen both in the execution time and in the memory consumption. Apart from the cost in performance, model fidelity for unknown inputs came also with a high price in loss of precision due to chained numerical computation, fact seen from the evaluation results of **Experiment 1** in the interpreter listing above. Choosing a better solver than `euler`, e.g. a Runge-Kutta solver, or even better, a symbolic solver, could improve both performance and precision, but that is out of the scope of this report. Another, rather surprising fact is that `osc2` performs slightly better than `osc1`, probably to the lack of tag calculus in the SY domain, i.e. tags are mainly passed untouched between interfaces, whereas calculations are performed in a synchronous reactive manner on values only.

Figure 3.11: Execution time and memory consumption for **Experiment 2**

To get a feeling of the cost complexity of the models for `osc1`, `osc2` and `osc4`, we plot the execution time and memory consumption during **Experiment 2** for three resolutions: 10 milliseconds, 50 milliseconds and 100 milliseconds. This way we can observe the trends in fig. 3.11, where the most notable one is the exponential growth in execution time for `osc4`. This can be explained recalling the so-called “time leaks” observed by Hudak et al., 2003 (among others), which occur due to the fact that lazy evaluation forces to recalculate all previous states of the ODE solver in order to evaluate the current sample. In functional reactive programming (FRP), a solution to time leaks is the concept of “continuation”, which is roughly what happens when the synchronous reactive FSM receives a new discrete event: it stores the current state to the ODE to be used in the future. Knowing this, let us see what happens when we double the number of discrete events within the input V_{DD} signal.

```

— | VDD input for performance testing, having twice more discrete events than 'vddTest'
vddTest1 = CT.signal [(0,λ_→2),(0.5,λ_→2),(1,λ_→1.5),(1.5,λ_→1.5),(2,λ_→1),(2.5,λ_→
1)] :: CT.Signal Rational

```

3.3 Conclusion

This project is still under development...

References

- Hudak, Paul et al. (2003). “Arrows, robots, and functional reactive programming”. In: *Lecture notes in computer science* 2638, pp. 159–187.
- Lee, Edward A. (2016). “Fundamental Limits of Cyber-Physical Systems Modeling”. In: *ACM Trans. Cyber-Phys. Syst.* 1.1, 3:1–3:26. ISSN: 2378-962X. DOI: [10.1145/2912149](https://doi.org/10.1145/2912149). URL: <http://doi.acm.org/10.1145/2912149>.
- Ungureanu, George, José E. G. de Medeiros, and Ingo Sander (2018). “Bridging discrete and continuous time with Atoms”. In: *2018 Design, Automation & Test in Europe Conference & Exhibition (DATE)*. IEEE.

Part II

API Documentation

This section is currently under construction. For the moment, please refer to online API documentation at <https://forsyde.github.io/forsyde-atom/>